


Examples



X2max
www.x2max.com

FileMaker Pro is a registered trademark of FileMaker, Inc.
© 1997-2002 by X2max Software. All rights reserved.

Table Of Contents

Examples

Introduction	3
Example 1	5
Example 2	9
Example 3	16
Example 4	22
Example 5	29
Example 6	38

Introduction

In the following the practical ways of using xmCHART are shown by way of selected examples. More extensive examples were purposely chosen to, on the one hand, show the versatility and power of xmCHART and, on the other hand, to illustrate the important interaction between FileMaker Pro scripts and xmCHART. By using FileMaker Pro scripts, it is possible to fully automate the creation of charts so that users can create expressive charts "at the touch of a button". All examples shown here are available as fully functional unlocked FileMaker Pro .FP3 and .FP5 files.

Integrating xmCHART into FileMaker Pro databases:

• FileMaker Pro fields:

In general, xmCHART requires three FileMaker Pro fields; advantageously, they are global fields which are denoted by the prefix "g" (global) in the following:

- gFunctions*: contains all function calls required for xmCHART.
- gChart*: contains the chart created by xmCHART.
- gError*: contains information on possible errors that may occur while creating the chart.

Additional so-called placeholders can be used to make the practical handling easier. To do this, another text global, for example with the name *gFunctionsWithPlaceholders*, is introduced. Just like the text variable *gFunctions*, it contains all function calls which are necessary for xmCHART, but instead of the "variable" data — i.e. the data which is normally taken from the database fields — placeholders are entered. Placeholders prove useful, e.g. for chart data, axis labels or legend texts. Placeholders are generally marked by pointed brackets "<>". For example:

```
ChartData(<gData>)  
AxisMajorTickLabelTexts(x;<gAxisLabels>)
```

A separate text global is set up for each placeholder, e.g. for <gData> a text global *gData*, for <gAxisLabels> a text global *gAxisLabels*, etc. The contents of these text globals are generally set up by

a FileMaker Pro script. Finally all placeholders are replaced by the actual values using another FileMaker Pro script. The working with placeholders is explained in the first three examples.

- *FileMaker Pro scripts:*

To draw a chart automatically at least one FileMaker Pro script is necessary which, for the most part, is divided into several subscripts to increase the clarity. The names of the subscripts always begin with an underscore, e.g. *_SubstitutePlaceholders*.

To familiarize yourself with xmCHART quickly and easily, the field and script names used above have been kept the same in all examples.

In the example files an effort was made to represent the essential steps, starting with the collection of values to be depicted down to the finished chart, in a clear and comprehensible manner. However, in a professional database solution, neither the scripts necessary to create the function calls nor the function calls themselves are generally visible for the user so that a chart can mostly be created "at the touch of a button". It is up to the developer of a database solution as to which options for creating a chart would be made available to the user.

As the FileMaker Pro plug-in API only allows the input and output of a text string, there is no cross-platform way to access database fields directly from xmCHART. In other words, all data necessary for creating a chart has to be copied to xmCHART via the `ChartData()` function. For this reason, the script needed to set up the `ChartData()` function usually holds a central position. Possible methods are shown in the following examples. The other function calls necessary for xmCHART can mostly be added per script by concatenating text strings.

PLEASE NOTE: *When entering text strings or font names in the Script Editor, it should be noted that the required double quotes must always be entered twice. For example:*

```
gFunctions & "TitleText(" "Chart 1" ")"
```

Example 1

Problem:

In the first example a series of values is represented in the form of a pie chart (Fig. 8). The individual slices of the pie should be labeled with the name of the category, the actual value and the percentage. The corresponding FileMaker Pro database is xmEXAMPLE1.FP3 or xmEXAMPLE1.FP5.

Solution:

FileMaker Pro fields: (Fig. 1)

In addition to the global fields *gFunctions*, *gChart*, *gError* and *gFunctions-WithPlaceholders* described previously in the introduction, the following fields are also necessary:

- The text field *Name* which contains the name of the category.
- The number field *Value* which contains the value of the category.
- The text globals *gData*, *gLabels* and *gFormat*. They correspond to the matching placeholders *<gData>*, *<gLabels>* and *<gFormat>* and serve to store the actual values.
- Optional: the global fields *gIsActive* and *gVersion*. If xmCHART is active, *gIsActive* contains the value "1", otherwise *gIsActive* is empty. *gVersion* contains the exact name of the presently active xmCHART version, e.g. "xmCHART 2.2".

<u>Field Name</u>	<u>Type</u>	<u>Options</u>
[1]	Calculation	Indexed, = 1
Name	Text	
Value	Number	
gFunctionsWithPlaceho...	Global	Text
gFunctions	Global	Text
gChart	Global	Container
gError	Global	Text
gVersion	Global	Text
gIsActive	Global	Text
gData	Global	Text
gLabels	Global	Text
gFormat	Global	Text

Fig. 1

FileMaker Pro scripts:

- FileMaker Pro script *SetUpChart*: (Fig. 2)

The script *SetUpChart* produces the desired chart and consists of several subscripts. As an option, it is possible to inquire whether the appropriate FileMaker Pro version is available (script: *_IsFMVersion4.0+*) and whether xmCHART is active (script: *_IsPluginActive*).

```
Freeze Window
Perform Script [Sub-scripts, "_IsFMVersion4.0+"]
Perform Script [Sub-scripts, "_IsPluginActive"]
Perform Script [Sub-scripts, "_SetUpData"]
Perform Script [Sub-scripts, "_SetUpLabels"]
Perform Script [Sub-scripts, "_SubstitutePlaceholders"]
Perform Script [Sub-scripts, "_DrawChart"]
```

Fig. 2

(1) _SetUpData: (Fig. 3)

The subscript *_SetUpData* takes the necessary data from the database, more precisely from the number field *Value*, and stores it in the form of a character string — the individual values are separated by spaces — in *gData*. This is done by using a loop over all records.

```
Set Field ["gData", ""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gData", "gData & Value & " "]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
```

Fig. 3

(2) _SetUpLabels: (Fig. 4)

The subscript *_SetUpLabels* takes the category names from the database and stores them in *gLabels*. The individual names are to be placed in double quotes and separated by semicolons. In addition, *gFormat* is added to every name. The text global *gFormat* defines the presentation of the numerical values: the first format specifier defines the output of the absolute values, the second format specifier the output of the percent values. Format specifiers are discussed in detail in *xmReference* and the various options for labeling pie charts in the *Styles* section of *xmTutorial*. Please note that when entering text or font names in the Script Editor the required double quotes must always be entered twice. For example:

```
gFunctions & "TitleText(" "Chart 1" ")"
```

```

Set Field ["gLabels", ""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gLabels", "gLabels & "" & Name & gFormat & "";""]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gLabels", "Left(gLabels; Length(gLabels)-1)"]

```

Fig. 4

(3) *_SubstitutePlaceholders*: (Fig. 5)

The subscript *_SubstitutePlaceholders* replaces the placeholders in *gFunctionsWithPlaceholders* (Fig. 6) with the actual values.

```

Set Field ["gFunctions", "Substitute(gFunctionsWithPlaceholders, "<gData>", gData)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gLabels>", gLabels)"]

```

Fig. 5

```

OpenDrawing(400;300)
    ChartData(<gData>)
    PieChart(oval+label+shadow)
    BorderStyle(all;0) // hide borders
    PieChartAuxLines() // labels with auxiliary lines
    LabelTexts(1;<gLabels>)
    LabelStyle(all;;;;center) // labels centered
    LabelBackground(all;lightYellow;1;1;2)
CloseDrawing()

```

Fig. 6

(4) *_DrawChart*: (Fig. 7)

In the subscript *_DrawChart* all functions included in *gFunctions* are sent to xmCHART by the external function `xmCH-DrawChart()`. If no error occurs when drawing the chart, the chart is then copied from the clipboard to the container field *gChart*.

```

Set Field ["gError", "External("xmCH-DrawChart", gFunctions)"]
If ["IsEmpty(gError)"]
    Paste ["gChart"]
End If

```

Fig. 7


Result:

Fig. 8

Variations:

The labels can be positioned inside the chart or split, i.e. both inside and outside the chart. In addition, a text can also be placed in the center of the pie chart. Details with examples can be found in *xmTutorial*, section *Charts*.

Example 2

Problem:

In the second example time-controlled data in the form of a so-called Gantt chart are to be represented (Fig. 18). To put it more precisely, the different periods of time for a series of tasks are to be plotted along a horizontal time axis. Depending on the period to be represented - whether it's days, weeks, months or years - an appropriate scaling should automatically be found. The different types of scaling can be tested by changing the periods of time (*from-to*) in the corresponding FileMaker Pro file xmEXAMPLE2.FP3 or xmEXAMPLE2.FP5.

Solution:**FileMaker Pro fields:** (Fig. 9)

In addition to the global fields *gFunctions*, *gChart*, *gError* and *gFunctions-WithPlaceholders* described previously in the introduction, the following fields are also necessary:

- The text field *Task* which contains the name of the task.
- The date fields *From1*, *To1*, *From2* and *To2*. A max. of two time periods can be defined for each task. In this concrete example the subdivision of each task was restricted to two time periods; however, in xmCHART each task can be divided into as many periods as desired.
- The global fields *gData*, *gXScaling*, *gXAxisLabels* and *gYAxisLabels* correspond to the matching placeholders *<gData>*, *<gXScaling>*, *<gXAxisLabels>* and *<gYAxisLabels>* and contain the actual values.
- The remaining date and number globals are necessary for calculating the different scaling periods and labels.
- Optional: the global fields *gIsActive* and *gVersion*. If xmCHART is active, *gIsActive* contains the value "1", otherwise *gIsActive* is empty. *gVersion* contains the exact name of the presently active xmCHART version, e.g. "xmCHART 2.2".

<u>Field Name</u>	<u>Type</u>	<u>Options</u>
[1]	Calculation	Indexed, = 1
Task	Text	
From1	Date	
To1	Date	
From2	Date	
To2	Date	
gFunctionsWithPlaceho...	Global	Text
gFunctions	Global	Text
gChart	Global	Container
gError	Global	Text
gData	Global	Text
gXScaling	Global	Text
gXAxisLabels	Global	Text
gYAxisLabels	Global	Text
gDate	Global	Date
gMinDate	Global	Date
gMaxDate	Global	Date
gStartDate	Global	Date
gEndDate	Global	Date
gNumOfYears	Global	Number
gNumOfMonths	Global	Number
gNumOfWeeks	Global	Number
gNumOfDays	Global	Number
gXScalingFactor	Global	Number
gCounter	Global	Number
gIsActive	Global	Text
gVersion	Global	Text

Fig. 9

FileMaker Pro scripts:

- FileMaker Pro script *SetUpChart*: (Fig. 10)

The script *SetUpChart* produces the desired chart and consists of several subscripts. As an option, it is possible to inquire whether the appropriate FileMaker Pro version is available (script: *_IsFMVersion4.0+*) and whether xmCHART is active (script: *_IsPluginActive*).

```

Freeze Window
Perform Script [Sub-scripts, "_IsFMVersion4.0+ "]
Perform Script [Sub-scripts, "_IsPluginActive "]
Perform Script [Sub-scripts, "_SetUpXScaling "]
Perform Script [Sub-scripts, "_SetUpXAxisLabels "]
Perform Script [Sub-scripts, "_SetUpYAxisLabels "]
Perform Script [Sub-scripts, "_SetUpData "]
Perform Script [Sub-scripts, "_SubstitutePlaceholders "]
Perform Script [Sub-scripts, "_DrawChart "]

```

Fig. 10

(1) _SetUpXScaling: (Fig. 11)

The subscript *_SetUpXScaling* calculates the appropriate scaling for the defined periods of time. First the subscript *__SetUpMinMaxDate* is called in which all records are searched for the earliest and latest date. They are stored in *gMinDate* and *gMaxDate*. In subscript *__CalculateRange* the number of years, months, weeks and days are calculated based on *gMinDate* and *gMaxDate*. Please note that four different methods of calculation are available for calculating the x-axis scale. They are dependent on the period of time to be represented, i.e. if the period of time to be represented is greater than 36 months, then an annual scale is used completed by a fine subdivision into quarters (script: *__Scaling: Years*). If the period of time is greater than 14 weeks and less than or equal to three years, a monthly scale is used (script: *__Scaling: Months*). If the period of time is greater than one week and less than or equal to 14 weeks, then a weekly scale is used completed by a fine subdivision into days (script: *__Scaling: Weeks*). For a period of time less than or equal to a week, a daily scale is used (script: *__Scaling: Days*).

```
Perform Script [Sub-scripts, "__SetUpMinMaxDate"]
Perform Script [Sub-scripts, "__CalculateRange"]
If ["gNumOfMonths > 36 "]
    Perform Script [Sub-scripts, "__Scaling: Years"]
Else
    If ["gNumOfWeeks > 14 "]
        Perform Script [Sub-scripts, "__Scaling: Months"]
    Else
        If ["gNumOfWeeks > 1 "]
            Perform Script [Sub-scripts, "__Scaling: Weeks"]
        Else
            Perform Script [Sub-scripts, "__Scaling: Days"]
        End If
    End If
End If
```

Fig. 11

(2) _SetUpXAxisLabels: (Fig. 12)

The subscript *_SetUpXAxisLabels* is set up in a way similar to the subscript *_SetUpXScaling*. Five types of labels are available for the four different scales. For an annual scale, i.e. the period of time is greater than 36 months, the beginning of each year is labeled, e.g. "Jan. 99" (script: *__AxisLabels: Years*). Two types of labels are available for a monthly scale: if the period of time is greater than 12 months and less than 36 months, the months are labeled using only the first letter (script: *__AxisLabels: Months1*); if the period of time is between 14 weeks and 12 months, the months are labeled using the first 3 letters (script: *__AxisLabels: Months2*). The year is added to the January months. For a period

of time greater than one week and less than 14 weeks, the beginning of each week (Monday) is labeled with the day and month (script: *__AxisLabels: Weeks*). If the period of time is only one week, the individual days are labeled with the weekday, day and month (script: *__AxisLabels: Days*). The individual scaling labels are to be placed in double quotes and separated by a semicolon. Please note that when entering texts in the Script Editor, the required double quotes must always be entered twice.

```
Set Field ["gXAxisLabels", ""]
If ["gNumOfMonths > 36"]
    Perform Script [Sub-scripts, "__AxisLabels: Years"]
Else
    If ["gNumOfMonths > 12"]
        Perform Script [Sub-scripts, "__AxisLabels: Months1"]
    Else
        If ["gNumOfMonths > 14"]
            Perform Script [Sub-scripts, "__AxisLabels: Months2"]
        Else
            If ["gNumOfMonths > 1"]
                Perform Script [Sub-scripts, "__AxisLabels: Weeks"]
            Else
                Perform Script [Sub-scripts, "__AxisLabels: Days"]
            End If
        End If
    End If
End If
End If
End If
```

Fig. 12

(3) *_SetUpYAxisLabels*: (Fig. 13)

The subscript *_SetUpYAxisLabels* takes the label texts for the y-axis from the database and stores them in *gYAxisLabels*. The individual names are to be placed in double quotes and separated by semicolons.

```
Set Field ["gYAxisLabels", ""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gYAxisLabels", "gYAxisLabels & "" & Task & "" ;"]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gYAxisLabels", "Left(gYAxisLabels; Length(gYAxisLabels)-1)"]
```

Fig. 13

(4) _SetUpData: (Fig. 14)

By using a loop over all records (one record corresponds to one task), the subscript *_SetUpData* takes the dates *From1*, *To1*, *From2* and *To2* from the database and stores them in the form of a string in *gData*. Only the numbers of days since *gStartDate* are stored. In addition, the numbers of days are adjusted to the scale used by multiplying with *gXScalingFactor*. Since each record corresponds to a data series, a semicolon must be added to the string after each record. In addition, the completeness of the data has to be checked. In other words, the period of time for each task must be clearly defined, otherwise it will be ignored.

```

Set Field ["gData", ""]
Go to Record/Request/Page [First]
Loop
  If ["not IsEmpty(From1) and not IsEmpty(To1)"]
    Set Field ["gData", "gData & NumToText((From1 - gStartDate) * gXScalingFactor) & " " &
    If ["not IsEmpty(From2) and not IsEmpty(To2)"]
      Set Field ["gData", "gData & NumToText((From2 - gStartDate) * gXScalingFactor) & " "
    Else
      Set Field ["gData", "gData & ";"]
    End If
  Else
    If ["not IsEmpty(From2) and not IsEmpty(To2)"]
      Set Field ["gData", "gData & NumToText((From2 - gStartDate) * gXScalingFactor) & " "
    Else
      Set Field ["gData", "gData & "0 0;"]
    End If
  End If
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gData", "Left(gData, Length(gData) - 1)"]

```

Fig. 14

(5) _SubstitutePlaceholders: (Fig. 15)

The subscript *_SubstitutePlaceholders* replaces the placeholders in *gFunctionsWithPlaceholders* (Fig. 16) with the actual values.

```

Set Field ["gFunctions", "Substitute(gFunctionsWithPlaceholders, "<gData>", gData)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gXScaling>", gXScaling)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gXAxisLabels>", gXAxisLabels)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gYAxisLabels>", gYAxisLabels)"]

```

Fig. 15

```

OpenDrawing(400;300)
  ChartData(<gData>)
  GanttChart(shadow)
  FillStyle(all;darkYellow)
  ShadowStyle(all;1)
  Scaling(x;linear;<gXScaling>)
  ScalingOptions(y;on) // y-scaling top to bottom
  // axes
  AxisMajorTickLabelTexts(x;<gXAxisLabels>)
  AxisMajorTickLabelTexts(y;<gYAxisLabels>)
  AxisLine(all;0)
  AxisMajorTicks(y;0)
  AxisMajorTicks(all;0)
  AxisMinorTicks(all;0)
  // grid
  MajorGridLineWidths(x;y;0) // hide horizontal grid
  MinorGridLineColors(all;all;lightGray)
  GridFrame(xy;1;gray)
CloseDrawing()

```

Fig. 16

(6) *_DrawChart*: (Fig. 17)

In the subscript *_DrawChart* all functions included in *gFunctions* are sent to xmCHART by the external function `xmCH-DrawChart()`. If no error occurs when drawing the chart, the chart is then copied from the clipboard to the container field *gChart*.

```

Set Field ["gError", "External("xmCH-DrawChart", gFunctions)"]
If ["IsEmpty(gError)"]
  Paste ["gChart"]
End If

```

Fig. 17


Result:

Fig. 18

Variations:

The four scalings used in this example can be refined even further; the same goes for the scaling labels. In addition, the bars can be labeled or rotated 90° so that the time axis runs from top to bottom instead of from left to right. Examples of this can be found in the *Charts* section of *xmTutorial*.

Example 3

Problem:

In the third example two charts, a bar chart and a line chart, are to be positioned over each other (Fig. 27). The corresponding FileMaker Pro file is xmEXAMPLE3.FP3 or xmEXAMPLE3.FP5. The following points are to be noted when overlaying charts:

- The charts, more precisely the areas enclosed by the grid, must cover each other exactly. This can be done by using the `OpenChart()` and `CloseChart()` functions. Please note that *isPlotArea=on* is set. The basic setup is:

```
OpenChart(left;top;width;height;isPlotArea)
//
// set up chart 1
//
CloseChart()

OpenChart(left;top;width;height;isPlotArea)
//
// set up chart 2
//
CloseChart()
```

- The scaling of both charts must be identical — exceptions are charts with two separately arranged y-axes. Since the individual overlayed charts are completely independent from each other and "do not know anything about each other", the scaling for both charts has to be set exactly the same either manually by using the `Scaling()` function or with a small trick: in this concrete example the minimum and maximum values are determined while copying the chart data from the database. Then these extreme values are added to both charts as a second data series. By doing so, it is guaranteed that both charts have the same y-scaling. The second data series is made invisible by using the appropriate style functions. For non-stacked bar charts the invisible second series can also be moved behind the first series by using *seriesGap=-100* so that the bars and labels under the x-axis match exactly.

- Furthermore, it is advantageous to display only the axes and grids of the first "bottom" chart and to suppress them on the second "top" chart.

Solution:

FileMaker Pro fields: (Fig. 19)

In addition to the global fields *gFunctions*, *gChart*, *gError* and *gFunctions-WithPlaceholders* described previously in the introduction, the following fields are also necessary:

- The text field *Name* which contains the labels for the x-axis.
- The number fields *BarChartValue* and *LineChartValue* which contain the respective chart data.
- The global fields *gBarChartValues*, *gLineChartValues*, *gXAxisLabels* and *gMinValue*. They correspond to the matching placeholders *<gBarChartValues>*, *<gLineChartValues>*, *<gXAxisLabels>* and *<gMinValue>* and contain the actual values. There is also the global field *gMaxValue*.
- Optional: the global fields *gIsActive* and *gVersion*. If xmCHART is active, *gIsActive* contains the value "1", otherwise *gIsActive* is empty. *gVersion* contains the exact name of the presently active xmCHART version, e.g. "xmCHART 2.2".

Field Name	Type	Options
[1]	Calculation	Indexed, = 1
Name	Text	
BarChartValue	Number	
LineChartValue	Number	
gFunctionsWithPlaceho...	Global	Text
gFunctions	Global	Text
gChart	Global	Container
gError	Global	Text
gVersion	Global	Text
gIsActive	Global	Text
gBarChartValues	Global	Text
gLineChartValues	Global	Text
gXAxisLabels	Global	Text
gMinValue	Global	Number
gMaxValue	Global	Number

Fig. 19

FileMaker Pro scripts:

- FileMaker Pro script *SetUpChart*: (Fig. 20)

The script *SetUpChart* produces the desired chart and consists of several subscripts. As an option, it is possible to inquire whether the appropriate FileMaker Pro version is available (script: *_IsFMVersion4.0+*) and whether xmCHART is active (script: *_IsPluginActive*).

```

Freeze Window
Perform Script [Sub-scripts, "_IsFMVersion4.0+"]
Perform Script [Sub-scripts, "_IsPluginActive"]
Perform Script [Sub-scripts, "_SetUpMinMaxValues"]
Perform Script [Sub-scripts, "_SetUpData"]
Perform Script [Sub-scripts, "_SetUpXAxisLabels"]
Perform Script [Sub-scripts, "_SubstitutePlaceholders"]
Perform Script [Sub-scripts, "_DrawChart"]

```

Fig. 20

(1) *_SetUpMinMaxValues*: (Fig. 21)

The subscript *_SetUpMinMaxValues* searches for the lowest and highest value in *BarChartValue* and *LineChartValue* and stores them in *gMinValue* and *gMaxValue*.

```

Set Field ["gMinValue", "0"]
Set Field ["gMaxValue", "0"]
Go to Record/Request/Page [First]
Loop
  If ["not IsEmpty(gBarChartValues)"]
    Set Field ["gMinValue", "If(BarChartValue < gMinValue; BarChartValue; gMinValue)"]
    Set Field ["gMaxValue", "If(BarChartValue > gMaxValue; BarChartValue; gMaxValue)"]
  End If
  If ["not IsEmpty(LineChartValue)"]
    Set Field ["gMinValue", "If(LineChartValue < gMinValue; LineChartValue; gMinValue)"]
    Set Field ["gMaxValue", "If(LineChartValue > gMaxValue; LineChartValue; gMaxValue)"]
  End If
  Go to Record/Request/Page [Exit after last, Next]
End Loop

```

Fig. 21

(2) *_SetUpData*: (Fig. 22)

The subscript *_SetUpData* takes the values necessary for drawing the bar chart and line chart from the database. The values are stored in the form of a character string — the individual numbers are separated by spaces. This is done by using a loop over all records. Possible missing values for the bar chart are replaced by zero. For the line chart they are substituted by a value less than *gMinValue*, for example *gMinValue-1*. If a minimum value is defined using *ChartDataLowerLimits(gMinValue)* — i.e. all values less than *gMinValue* are ignored — the line is discontinuous at the missing points. More information and examples can be found in the *Data* section of *xmTutorial*.

Finally, the extreme values *gMinValue* and *gMaxValue* are added as the second data series. By doing so, it is guaranteed that both charts have the same y-scaling.

```

Set Field ["gBarChartValues", ""]
Set Field ["gLineChartValues", ""]
Go to Record/Request/Page [First]
Loop
  If ["IsEmpty(BarChartValue)"]
    Set Field ["gBarChartValues", "gBarChartValues & "0 "]
  Else
    Set Field ["gBarChartValues", "gBarChartValues & BarChartValue & " "]
  End If
  If ["IsEmpty(LineChartValue)"]
    Set Field ["gLineChartValues", "gLineChartValues & NumToText(gMinValue-1) & " "]
  Else
    Set Field ["gLineChartValues", "gLineChartValues & LineChartValue & " "]
  End If
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gBarChartValues", "gBarChartValues & ";" & gMinValue & " " & gMaxValue"]
Set Field ["gLineChartValues", "gLineChartValues & ";" & gMinValue & " " & gMaxValue"]

```

Fig. 22

(3) *_SetUpXAxisLabels*: (Fig. 23)

The subscript *_SetUpXAxisLabels* takes the label texts for the x-axis from the database and stores them in the form of a character string in *gXAxisLabels*. The individual names are to be placed in double quotes and separated by semicolons. Please note that when entering texts in the Script Editor the required double quotes must always be entered twice.

```

Set Field ["gXAxisLabels", ""]
Go to Record/Request/Page [First]
Loop
  Set Field ["gXAxisLabels", "gXAxisLabels & """" & Name & """;"]
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gXAxisLabels", "Left(gXAxisLabels; Length(gXAxisLabels)-1)"]

```

Fig. 23

(4) *_SubstitutePlaceholders*: (Fig. 24)

The subscript *_SubstitutePlaceholders* replaces the placeholders in *gFunctionsWithPlaceholders* (Fig. 25) with the actual values.

```

Set Field ["gFunctions", "Substitute(gFunctionsWithPlaceholders, "<gMinValue>", gMinValue)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gBarChartValues>", gBarChartValues)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gLineChartValues>", gLineChartValues)"]
Set Field ["gFunctions", "Substitute(gFunctions, "<gXAxisLabels>", gXAxisLabels)"]

```

Fig. 24

```

OpenDrawing(400;300)
// BAR CHART
OpenChart(40;20;340;250;on)
    ChartData(<gBarChartValues>)
    BarChart(label;;-100)
    AxisMajorTicks(all;0) // hide tick marks
    AxisMajorTickLabelTexts(x;<gXAxisLabels>)
    // grid
    MajorGridLineWidths(y;x;0) // hide vertical grid
    MajorGridLineColors(all;all;lightGray)
    MajorGridStripeColors(x;y;240 240 240;250 250 250)
    GridFrame()
    // hide 2nd series
    BorderStyle(2;none)
    FillStyle(2;;transparent)
    LabelTexts(2;" ")
CloseChart()

// LINE CHART
OpenChart(40;20;340;250;on)
    ChartData(<gLineChartValues>)
    ChartDataLowerLimits(<gMinValue>)
    LineChart(label;on)
    LineStyle(1;poly;2;darkYellow)
    LabelBackground(1;lightYellow)
    LabelOptions(1;centerCenter)
    AxisOptions(all;none) // hide axes
    GridLocation(;none) // hide grid
    // hide 2nd series
    LineStyle(2;none)
    LabelTexts(2;" ")
CloseChart()
CloseDrawing()

```

Fig. 25

(5) *_DrawChart*: (Fig. 26)

In the subscript *_DrawChart* all functions included in *gFunctions* are sent to xmCHART by the external function *xmCH-DrawChart()*. If no error occurs when drawing the chart, the chart is then copied from the clipboard to the container field *gChart*.

```
Set Field ["gError", "External("xmCH-DrawChart", gFunctions)"]
If ["IsEmpty(gError)"]
    Paste ["gChart"]
End If
```

Fig. 26


Result:

Fig. 27

Variations:

As an option, a separate scaling can be used for bar and line charts. For example, the y-axis of the bar chart is drawn on the left (default) and the y-axis of the line chart on the right side - the y-axis is shifted to the right by using the *AxisOptions(y; on)* function. An example of this can be found in *xmGALLERY.FP3* or *xmGALLERY.FP5*.

Example 4

In the fourth example, the field contents and their sums are to be evaluated graphically. The corresponding FileMaker Pro file is xmEXAMPLE4.FP3 or xmEXAMPLE4.FP5.

Problem:

Let's assume you want to graphically evaluate the turnover of your company. On the one hand, the turnover should be evaluated quarterly; on the other hand, it should also be categorized according to employees and represented in the form of a stacked bar chart.

For this purpose, you maintain an employee database in which the turnover of the individual employees is recorded per quarter. One record is set up for each employee with the fields *Employee_Name*, *Turnover_Q1*, *Turnover_Q2*, *Turnover_Q3*, *Turnover_Q4*, as well as the summary fields *Total_Turnover_Q1*, *Total_Turnover_Q2*, *Total_Turnover_Q3*, *Total_Turnover_Q4* and the formula field *Employee_Turnover*, which calculates the total turnover of an employee within one year, followed by the evaluation field *Total_Turnover*, which contains the sum of all turnover within one year.

Solution:

FileMaker Pro fields: (Fig. 28)

In addition to the data fields described previously, the following fields are also necessary:

- The three "xmCHART Globals" *gFunctions*, *gChart* and *gError*.
- Optional: the global fields *gIsActive* and *gVersion*. If xmCHART is active, *gIsActive* contains the value "1", otherwise *gIsActive* is empty. *gVersion* contains the exact name of the presently active xmCHART version, e.g. "xmCHART 2.2".

<u>Field Name</u>	<u>Type</u>	<u>Options</u>
Employee_Name	Text	
Turnover_Q1	Number	
Turnover_Q2	Number	
Turnover_Q3	Number	
Turnover_Q4	Number	
Total_Turnover_Q1	Summary	= Total of Turnover_Q1
Total_Turnover_Q2	Summary	= Total of Turnover_Q2
Total_Turnover_Q3	Summary	= Total of Turnover_Q3
Total_Turnover_Q4	Summary	= Total of Turnover_Q4
Employee_Turnover	Calculation	= Turnover_Q1 + Turnover_Q2 + ...
Total_Turnover	Summary	= Total of Employee_Turnover
gFunctions	Global	Text
gChart	Global	Container
gError	Global	Text
gVersion	Global	Text
gIsActive	Global	Text

Fig. 28

Turnover per quarter:

- FileMaker Pro script *SetUpFunctions_TurnoverPerQuarter*:

The script *SetUpFunctions_TurnoverPerQuarter* is made up of three sub-scripts: (Fig. 29)

```

Freeze Window
Set Field ["gFunctions", "OpenDrawing(420;300)¶"]
Perform Script [Sub-scripts, "_ChartData_TurnOverPerQuarter"]
Perform Script [Sub-scripts, "_Functions_TurnOverPerQuarter"]
Perform Script [Sub-scripts, "_Legendtexts_TurnOverPerQuarter"]

```

Fig. 29

(1) _ChartData_TurnoverPerQuarter:

This subscript sets up the `ChartData()` function. In order to determine the turnover per quarter, the employee turnover is added to `ChartData()` by looping all records. (see Fig. 30)

```

Set Field ["gFunctions", "gFunctions & "ChartData(""]
Go to Record/Request/Page [First]
Loop
  Set Field ["gFunctions", "gFunctions & Turnover_Q1 & " " & Turnover_Q2 & ...
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "Left(gFunctions;Length(gFunctions) - 1) & ")¶"]

```

Fig. 30

(2) _Functions_TurnoverPerQuarter:

Other necessary function calls in addition to `ChartData()` are added to *gFunctions* as represented in Fig. 31.

```
gFunctions &
"BarChart(stacked)" &
"AxisMajorTickLabelTexts(x;"Q1";"Q2";"Q3";"Q4")" &
"TitleText("Turnover per Quarter")" &
"AddText(320;250;"Total Turnover:\n" & Total_Turnover & "";;;bold)"
```

Fig. 31

(3) _Legndtexts_TurnoverPerQuarter:

Since the legend should contain both the names of the employees and their turnover, the `LegendTexts()` function is set up dynamically. That means the names of the employees and all of their turnover are added to the `LegendTexts()` function by looping all records. (Fig. 32)

```
Set Field ["gFunctions", "gFunctions & "LegendTexts(""]
Go to Record/Request/Page [First]
Loop
  Set Field ["gFunctions", "gFunctions & "" & Employee_Name & "(" & Emplo...
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "Left(gFunctions;Length(gFunctions) - 1) & ")"]
```

Fig. 32

- FileMaker Pro field *gFunctions*:

After running the script *SetUpFunctions_TurnoverPerQuarter*, the global field *gFunctions* contains the following function calls for xmCHART. (see Fig. 33)

```
OpenDrawing(420;300)
ChartData(32500 62400 49700 42100; 12300 32500 53200;
23900 52300 53400 45200;30100 58200 56100 0;
0 0 22000 0)
BarChart(stacked)
AxisMajorTickLabelTexts(x;"Q1";"Q2";"Q3";"Q4")
TitleText("Turnover per Quarter")
AddText(320;250;"Total Turnover:\n625900";;;bold)
LegendTexts("Empl. 1 (186700)";"Empl. 2 (98000)";
"Empl. 3 (174800)";"Empl. 4 (144400)";"Empl. 5 (22000)")
```

Fig. 33

- FileMaker Pro field *gChart*:

The chart created by xmCHART and then stored in *gChart* is represented in Fig. 34.




Fig. 34

Turnover per employee:

- FileMaker Pro script *SetUpFunctions_TurnoverPerEmployee*:

The script *SetUpFunctions_TurnoverPerEmployee* is made up of four sub-scripts: (Fig. 35)

```

Freeze Window
Set Field ["gFunctions", "OpenDrawing(420;300)¶"]
Perform Script [Sub-scripts, "_ChartData_TurnOverPerEmployee"]
Perform Script [Sub-scripts, "_Functions_TurnOverPerEmployee"]
Perform Script [Sub-scripts, "_Legendtexts_TurnOverPerEmployee"]
Perform Script [Sub-scripts, "_AxisTickLabels_TurnOverPerQuar..."]

```

Fig. 35

(1) *_ChartData_TurnoverPerEmployee*:

The *ChartData()* function is set up by this subscript. In order to determine the turnover per employee, the turnover is added to *ChartData()* quarterly by looping all employees. That means a total of four data series (one series per quarter) are produced. (see Fig. 36)

```

Set Field ["gFunctions", "gFunctions & "ChartData(""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gFunctions", "gFunctions & Turnover_Q1 & " "]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "gFunctions & "; ""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gFunctions", "gFunctions & Turnover_Q2 & " "]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "gFunctions & "; ""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gFunctions", "gFunctions & Turnover_Q3 & " "]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "gFunctions & "; ""]
Go to Record/Request/Page [First]
Loop
    Set Field ["gFunctions", "gFunctions & Turnover_Q4 & " "]
    Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "Left(gFunctions;Length(gFunctions) - 1) & ")¶"]

```

Fig. 36

(2) _Functions_TurnoverPerEmployee:

Other necessary function calls in addition to ChartData() are added to *gFunctions* as represented in Fig. 37.

```

gFunctions &
"BarChart(stacked)¶" &
"TitleText("Turnover per Employee")¶" &
"AddText(320;250;"Total Turnover:\n" & Total_Turnover & "";;;bold)¶"

```

Fig. 37

(3) _Legendtexts_TurnoverPerEmployee:

The LegendTexts() function is put together by concatenating text constants and field contents like in Fig. 38.

```

gFunctions & "LegendTexts(
""1.Quarter (" & Total_Turnover_Q1 & ")"";
""2.Quarter (" & Total_Turnover_Q2 & ")"";
""3.Quarter (" & Total_Turnover_Q3 & ")"";
""4.Quarter (" & Total_Turnover_Q4 & ")""
)¶"

```

Fig. 38

(4) *_AxisTickLabels_TurnoverPerEmployee*:

Since the x-axis tick mark labels should contain the names of the employees, the `AxisMajorTickLabelTexts()` function is set up dynamically. That means the names of the employees are added to the `AxisMajorTickLabelTexts()` function by looping all employees. (Fig. 39)

```
Set Field ["gFunctions", "gFunctions & "AxisMajorTickLabelTexts(x;"]
Go to Record/Request/Page [First]
Loop
  Set Field ["gFunctions", "gFunctions & "" & Employee_Name & "";""]
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "Left(gFunctions,Length(gFunctions) - 1) & "¶"]
```

Fig. 39

- FileMaker Pro field *gFunctions*:

After running the script *SetUpFunctions_TurnoverPerEmployee*, the field *gFunctions* contains the following function calls for xmCHART. (Fig. 40)

```
OpenDrawing(420;300)
ChartData(32500 23900 30100 0 ; 62400 12300 52300
58200 0 ; 49700 32500 53400 56100 22000 ;
42100 53200 45200 0 0)
BarChart(stacked)
TitleText("Turnover per Employee")
AddText(320;250;"Total Turnover:\n625900";;bold)
LegendTexts( "1.Quarter (86500)"; "2.Quarter (185200)";
"3.Quarter (213700)"; "4.Quarter (140500)" )
AxisMajorTickLabelTexts(x;"Empl. 1";"Empl. 2";
"Empl. 3";"Empl. 4";"Empl. 5")
```

Fig. 40

- FileMaker Pro field *gChart*:

The chart created by xmCHART and then stored in *gChart* is represented in Fig. 41.




Fig. 41

Example 5

In the fifth example the field contents are to be evaluated graphically according to the frequency. The corresponding FileMaker Pro database is xmEXAMPLE5.FP3 or xmEXAMPLE5.FP5.

Problem:

Let's assume you have a website and you would like an overview of the distribution of visitor frequency within one day and within one week. In addition, you want to categorize the visitors to your website according to their native countries.

For this purpose, you have set up a FileMaker Pro file with three fields, *Hour*, *Weekday* and *Country*. You have taken the corresponding data from the access protocol file of your website. The number field *Hour* contains a value between 0 and 23 – more detailed information on minutes and seconds do not appear. The number field *Weekday* contains a number between 1 (Sunday) and 7 (Saturday) and the text field *Country* contains a country abbreviation consisting of two characters.

Solution:

FileMaker Pro fields: (Fig. 42)

In addition to the three previously described data fields, *Hour*, *Weekday* and *Country*, the following other fields are also necessary:

- The global fields *gCountry*, *gLabel* and *gCounter*.

These globals are necessary to set up the functions `ChartData()` and `AxisMajorTickLabelTexts()`.

- The three "xmCHART Globals" *gFunctions*, *gChart* and *gError*.
- Optional: the global fields *gIsActive* and *gVersion*. If xmCHART is active, *gIsActive* contains the value "1", otherwise *gIsActive* is empty. *gVersion* contains the exact name of the presently active xmCHART version, e.g. "xmCHART 2.2".

<u>Field Name</u>	<u>Type</u>	<u>Options</u>
Hour	Number	Indexed
Weekday	Number	Indexed
Country	Text	
gCountry	Global	Text
gLabel	Global	Text
gCounter	Global	Number
gFunctions	Global	Text
gChart	Global	Container
gError	Global	Text
gVersion	Global	Text
gIsActive	Global	Text

Fig. 42

Frequency distribution over 24 hours:

- FileMaker Pro script *SetUpFunctions_24Hours*:

The script *SetUpFunctions_24Hours* is made up of two subscripts: (see Fig. 43)

```

Freeze Window
Set Field ["gFunctions", "OpenDrawing(420;300)¶"]
Perform Script [Sub-scripts, "_ChartData_24Hours"]
Perform Script [Sub-scripts, "_Functions_24Hours"]

```

Fig. 43

(1) _ChartData_24Hours:

The `ChartData()` function is set up by this subscript. In order to determine the number of visitors per hour, records with *Hour*=0, *Hour*=1, *Hour*=2 up to *Hour*=23 are searched for via a loop and the number of records found is added to `ChartData()`. If no visitors are found within a certain hour interval, it should be noted that the resulting error (error number 401 – no records match this request) must be handled by means of the FileMaker Pro script step *Set Error Capture [On]* and by adding 0 to `ChartData()`. (see Fig. 44)

```

Set Error Capture [On]
Set Field ["gFunctions", "gFunctions & "ChartData(""]
Set Field ["gCounter", "0"]
Loop
  Enter Find Mode []
  Set Field ["Hour", "gCounter"]
  Perform Find []
  If ["Status(CurrentError) = 401"]
    Set Field ["gFunctions", "gFunctions & "0 "]
  Else
    Set Field ["gFunctions", "gFunctions & Status(CurrentFoundCount) & " "]
  End If
  Exit Loop If ["gCounter = 23"]
  Set Field ["gCounter", "gCounter + 1"]
End Loop
Set Field ["gFunctions", "Left(gFunctions;Length(gFunctions) - 1) & "␣"]
Set Error Capture [Off]

```

Fig. 44

(2) *_Functions_24Hours*:

Other necessary function calls in addition to `ChartData()` are added to *gFunctions* as represented in Fig. 45.

```

gFunctions &
"BarChart()␣" &
"PictureStyle(1;resource;""8""␣" &
"TitleText("""Frequency distribution over 24 hours""")␣" &
"TitleBackground()␣" &
"GridLocation(;none)␣" &
"AxisMajorTicks(y;3;;;out)␣" &
"AxisMajorTicks(x;0)␣"

```

Fig. 45

- FileMaker Pro field *gFunctions*:

After running the script *SetUpFunctions_24Hours*, the global field *gFunctions* contains the following function calls: (see Fig. 46)

```

OpenDrawing(420;300)
ChartData(65 46 37 33 24 33 36 47 59 42 41 47 55 61
62 76 63 68 70 57 62 73 65 66)
BarChart()
PictureStyle(1;resource;"8")
TitleText("Frequency distribution over 24 hours")
TitleBackground()
GridLocation(;none)
AxisMajorTicks(y;3;;;out)
AxisMajorTicks(x;0)

```

Fig. 46

- FileMaker Pro field *gChart*:

The chart created by xmCHART and then stored in *gChart* is represented in Fig. 47.




Fig. 47

Frequency distribution over weekdays:

- FileMaker Pro script *SetUpFunctions_Weekdays*:

The script *SetUpFunctions_Weekdays* is also made up of two subscripts: (see Fig. 48)

```
Freeze Window
Set Field ["gFunctions", "OpenDrawing(420;300)¶"]
Perform Script [Sub-scripts, "_ChartData_Weekdays"]
Perform Script [Sub-scripts, "_Functions_Weekdays"]
```

Fig. 48

(1) _ChartData_Weekdays:

The *ChartData()* function is set up by this subscript. In order to determine the number of visitors per weekday, records with *Weekday=1*, *Weekday=2* up to *Weekday=7* are searched for via a loop and the number of records found is added to *ChartData()*. If no visitors are found within a day, it should be noted that the resulting error (error number 401 – no records match this request) must be handled by means of the FileMaker Pro script step *Set Error Capture [On]* and by adding 0 to *ChartData()*. (Fig. 49)

```
Set Error Capture [On]
Set Field ["gFunctions", "gFunctions & "ChartData(""]
Set Field ["gCounter", "1"]
Loop
  Enter Find Mode []
  Set Field ["Weekday", "gCounter"]
  Perform Find []
  If ["Status(CurrentError) = 401"]
    Set Field ["gFunctions", "gFunctions & "0 ""]
  Else
    Set Field ["gFunctions", "gFunctions & Status(CurrentFoundCount) & " ""]
  End If
  Exit Loop If ["gCounter = 7"]
  Set Field ["gCounter", "gCounter + 1"]
End Loop
Set Field ["gFunctions", "Left(gFunctions;Length(gFunctions) - 1) & "¶"]
```

Fig. 49

(2) _Functions_Weekdays:

Other necessary function calls in addition to *ChartData()* are added to *gFunctions* as represented in Fig. 50.

```

gFunctions &
"BarChart(label)" &
"TitleText("Frequency distribution over weekdays")" &
"TitleBackground(;transparent)" &
"Background(;119;0)" &
"AxisMajorTicks(y;3;;;out)" &
"AxisMajorTicks(x;0)" &
"AxisMajorTickLabelTexts(x;"Sun";"Mon";"Tue";
"Wed";"Thu";"Fri";"Sat")"

```

Fig. 50

- FileMaker Pro field *gFunctions*:

After running the script *SetUpFunctions_Weekdays*, the global field *gFunctions* contains the following function calls: (see Fig. 51)

```

OpenDrawing(420;300)
ChartData(118 108 222 236 209 214 181)
BarChart(label)
TitleText("Frequency distribution over weekdays")
TitleBackground(;transparent)
Background(;119;0)
AxisMajorTicks(y;3;;;out)
AxisMajorTicks(x;0)
AxisMajorTickLabelTexts(x;"Sun";"Mon";"Tue";
"Wed";"Thu";"Fri";"Sat")

```

Fig. 51

- FileMaker Pro field *gChart*:

The chart created by xmCHART and then stored in *gChart* is represented in Fig. 52.




Fig. 52

Division according to countries:

- FileMaker Pro script *SetUpFunctions_Countries*:

The script *SetUpFunctions_Countries* is also made up of two subscripts:
(Fig. 53)

```
Freeze Window
Set Field ["gFunctions", "OpenDrawing(420;300)¶"]
Perform Script [Sub-scripts, "_ChartData_Countries"]
Perform Script [Sub-scripts, "_Functions_Countries"]
```

Fig. 53

(1) *_ChartData_Countries*:

The functions *ChartData()* and *AxisMajorTickLabelTexts()* are set up by this subscript. In order to classify the visitors according to their native countries, the records are first sorted according to country abbreviations and then the frequencies of the individual country abbreviations are totaled and added to *ChartData()* by looping all records. In addition, the country abbreviations and their frequencies are stored in the global field *gLabel*. (Fig. 54)

```

Set Field ["gFunctions", "gFunctions & "ChartData(""]
Show All Records
Sort [Restore, No dialog]
Go to Record/Request/Page [First]
Set Field ["gLabel", ""AxisMajorTickLabelStyle(y;""Courier"";10)¶"]
Set Field ["gLabel", ""AxisMajorTickLabelTexts(y;"""]
Set Field ["gCountry", "Country"]
Set Field ["gCounter", "1"]
Loop
  Go to Record/Request/Page [Exit after last, Next]
  If ["Country = gCountry"]
    Set Field ["gCounter", "gCounter + 1"]
  Else
    Set Field ["gFunctions", "gFunctions & gCounter & " "]
    Set Field ["gLabel", "gLabel & ""(" & gCounter & ") " & gCountry & "";""]
    Set Field ["gCountry", "Country"]
    Set Field ["gCounter", "1"]
  End If
End Loop
Set Field ["gFunctions", "gFunctions & gCounter & " "]
Set Field ["gLabel", "gLabel & ""(" & gCounter & ") " & gCountry & """""]
Set Field ["gCountry", "Country"]
Set Field ["gCounter", "1"]
Set Field ["gFunctions", "Left(gFunctions, Length(gFunctions) - 1) & ")¶"]
Set Field ["gFunctions", "gFunctions & gLabel & ")¶"]

```

Fig. 54

(2) *_Functions_Countries:*

Other necessary function calls in addition to `ChartData()` and `AxisMajorTickLabelTexts()` are added to *gFunctions* as represented in Fig. 55.

```

gFunctions &
"BarChart(horizontal)¶" &
"TitleText( ""Division according to countries"" )¶" &
"TitleBackground(,transparent)¶" &
"BackgroundPict(resource;""29"" )¶" &
"AxisMajorTicks(x;3;;;out)¶" &
"AxisMajorTicks(y;0)¶" &
"GridLocation(,none)¶"

```

Fig. 55

• FileMaker Pro field *gFunctions*:

After running the script *SetUpFunctions_Countries*, the global field *gFunctions* contains the following function calls: (see Fig. 56)

```

OpenDrawing(420;300)
ChartData(6 34 76 23 117 54 148 16 23 6 46 64 75 44 8
48 8 7 47 438)
AxisMajorTickLabelTexts(y;"(6) AR";"(34) AT";
"(76) AU";"(23) BE";"(117) CA";"(54) CH";"(148) DE";
"(16) DK";"(23) ES";"(6) FI";"(46) FR";"(64) GB";
"(75) IT";"(44) JP";"(8) MX";"(48) NL";"(8) NO";
"(7) NZ";"(47) SE";"(438) US")
BarChart(horizontal)
TitleText("Division according to countries")
TitleBackground(;transparent)
BackgroundPict(resource;"29")
AxisMajorTicks(x;3;;;out)
AxisMajorTicks(y;0)
GridLocation(;none)

```

Fig. 56

- FileMaker Pro field *gChart*:

The chart created by xmCHART and then stored in *gChart* is represented in Fig. 57.




Fig. 57

Example 6

In the sixth example time-dependent data is evaluated graphically. The corresponding FileMaker Pro file is xmEXAMPLE6.FP3 or xmEXAMPLE6.FP5.

Problem:

Let's assume you record three readings in a 14-day interval which are to be represented graphically by a line chart. Furthermore, the series of readings are to begin at different times.

For this purpose, you set up a FileMaker Pro database with four fields, *Date*, *Series1*, *Series2* and *Series3*. A new record is set up for each day of reading. In this example a 2-dimensional line chart is used which is quite versatile. Thus, for example, data series can begin at different times or, furthermore, it would also be possible to use reading intervals of varying lengths.

FileMaker Pro fields: (Fig. 58)

In addition to the data fields described previously, the following fields are also necessary:

- The three "xmCHART Globals" *gFunctions*, *gChart* and *gError*.
- Optional: the global fields *gIsActive* and *gVersion*. If xmCHART is active, *gIsActive* contains the value "1", otherwise *gIsActive* is empty. *gVersion* contains the exact name of the presently active xmCHART version, e.g. "xmCHART 2.2".

<u>Field Name</u>	<u>Type</u>	<u>Options</u>
Date	Date	
Series1	Number	
Series2	Number	
Series3	Number	
gFunctions	Global	Text
gChart	Global	Container
gError	Global	Text
gVersion	Global	Text
gIsActive	Global	Text

Fig. 58

FileMaker Pro scripts:• FileMaker Pro script *SetUpFunctions*:

The script *SetUpFunctions* is made up of two subscripts: (Fig. 59)

```
Freeze Window
Set Field ["gFunctions", "OpenDrawing(420;300)¶"]
Perform Script [Sub-scripts, "_ChartData"]
Perform Script [Sub-scripts, "_Functions"]
```

Fig. 59

(1) *_ChartData*:

In order to make this process more clear, the script *_ChartData* calls up three other subscripts in which the *ChartData()* function is set up. (see Fig. 60)

```
Set Field ["gFunctions", "gFunctions & "ChartData(""]
Perform Script [Sub-scripts, "__Series1"]
Perform Script [Sub-scripts, "__Series2"]
Perform Script [Sub-scripts, "__Series3"]
Set Field ["gFunctions", "Left(gFunctions; Length(gFunctions) - 2) & ")¶"]
```

Fig. 60

The necessary values for representing the first series of readings are extracted by the script *__Series1* shown in Fig. 61. First, the reading dates or, more precisely, the days of the year are added to function *ChartData()* by looping all records. Then the corresponding readings are attached to *ChartData()* by looping all records again.

```
Go to Record/Request/Page [First]
Loop
  If ["not IsEmpty(Series1)"]
    Set Field ["gFunctions", "gFunctions & DayofYear(Date) & " "]
  End If
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "gFunctions & "; ""]
Go to Record/Request/Page [First]
Loop
  If ["not IsEmpty(Series1)"]
    Set Field ["gFunctions", "gFunctions & Series1 & " "]
  End If
  Go to Record/Request/Page [Exit after last, Next]
End Loop
Set Field ["gFunctions", "gFunctions & "; ""]
```

Fig. 61

(2) _Functions:

Other necessary function calls in addition to `ChartData()` are added to *gFunctions* as represented in Fig. 62.

```
gFunctions &
"LineChart2D()" &
"LineStyle(all;;2)" &
"LegendStyle()" &
"Background (240 240 255;;0)" &
"Scaling(y;linear;400;600;4;5)" &
"AxisMinorTicks(y;3;;;out)" &
"Scaling(x;linear;1;365;12;1)" &
"AxisMajorTickLabelTexts(x;""1 Jan"";""\n1 Feb"";
""1 Mar"";""\n1 Apr"";""1 May"";""\n1 Jun"";
""1 Jul"";""\n1 Aug"";""1 Sep"";""\n1 Oct"";
""1 Nov"";""\n1 Dec"";""31 Dec"")"
```

Fig. 62

- FileMaker Pro field *gFunctions*:

After running the script *SetUpFunctions*, the global field *gFunctions* contains the following function calls: (Fig. 63)

```
OpenDrawing(420;300)
ChartData(1 15 29 43 57 71 85 99 113 127 141 155 169
183 197 211 225 239 253 267 281 295 ; 428 531 493 473
472 473 442 443 467 454 442 475 485 472 489 499 492
512 510 501 519 532 ; 71 85 99 113 127 141 155 169 183
197 211 225 239 253 267 281 295 ; 540 572 567 557 554
563 561 571 572 573 564 542 555 567 572 575 577 ; 113
127 141 155 169 183 197 211 225 239 253 267 281 295 ;
500 505 510 520 525 521 510 512 496 490 489 483 489
479 )
LineChart2D()
LineStyle(all;;2)
LegendStyle()
Background (240 240 255;;0)
Scaling(y;linear;400;600;4;5)
AxisMinorTicks(y;3;;;out)
Scaling(x;linear;1;365;12;1)
AxisMajorTickLabelTexts(x;"1 Jan";"\n1 Feb"; "1 Mar";
"\n1 Apr";"1 May";"\n1 Jun"; "1 Jul";"\n1 Aug";"1 Sep";
"\n1 Oct"; "1 Nov";"\n1 Dec";"31 Dec")
```

Fig. 63

- FileMaker Pro field *gChart*:

The chart created by xmCHART and then stored in *gChart* is represented in Fig. 64.




Fig. 64